
;******************************************************************************

; HD44780 LCD Routines

;******************************************************************************

POR

;POR according to Hitachi datasheet.  NB: RB<0:3> are mapped to display lines 

;DB<4:7>, in order.  Use PutNib for the POR unitl 4-bit mode is set.   A swapf

;is not needed in PutNib as the lower 4 bits of PORTB are sent first and only

;those bits are sent. 

     DelayCy   (100*msecs)

     movlw     0x03           

     call      PutNib

     DelayCy   (5*msecs)

     movlw     0x03

     call      PutNib

     DelayCy   (1*msecs)

     movlw     0x03

     call      PutNib          

     movlw     0x02           ;Function set: read as 0x20 

     call      PutNib         ;4-bit mode is now set, N & F not defined yet

;LCD is now in 4-bitmode and 8-bit instructions and data are sent as two nibbles.

     movlw     0x28           ;Function set: 8 bits sent 4-bit mode 

     call      PutCmd         ;N & F are now defined

     movlw     0x08           ;Display on/off control: display off

     call      PutCmd         

     DelayCy   (150*msecs)    ;blinks screen (optional)

     DelayCy   (150*msecs)    ;blinks screen (optional)

     movlw     0x01           ;Clear display: clears display

     call      PutCmd         

     DelayCy   (2*msecs)

    movlw  0x06           ;Entry mode set: increment address (cursor)

  call   PutCmd         

  movlw  0x0C           ;Display on/off: display on/cursor off/blink off

  call   PutCmd

    

  return

;******************************************************************************

;LCD Subroutines

;******************************************************************************

PutNib

     movwf     PORTB          ;PORTB is bits <0:3> i.e., lower nibble only 

     bsf       E              ;this routine only send the low nibble

     bcf       E

;     call      WNB

     DelayCy   (50*usecs)

     return

PutCmd    

     bcf       RS

     goto      $+2   

PutDat     

bsf       RS

     movwf     PORTB

     swapf     PORTB,f

;Since PORTB is low nibble, swap low and high to put high in lower nibble and

;send first.

     bsf       E                 

     bcf       E      

     movwf     PORTB               

     bsf       E

     bcf       E

 ;    call      WNB

     DelayCy   (50*usecs)      

     return

